The Missing SEX Problem

Nick Holford, July 2006
Anthe Zandvliet wrote:

>

> Dear NMusers,

>

> A few years ago, Nick Holford wrote the following to NMusers about

> handling of missing SEX data:

>

> "... An alternative, more elegant approach, is to treat SEX as another

> DV. This is a bit trickier as it

> requires a LIKELIHOOD model that allows you to estimate continuous and

> categorical data at the

> same time. The missing SEX values are then predicted from the parameter

> describing the probability

> of being female just like you can predict DV values at times when you

> have no observations."

> http://huxley.phor.com/nonmem/nm/99jul302002.html

>

> I am trying to implement this into an example control stream (see

> below). Thus far, I have managed

> to simultaneously estimate continuous and categorical data. Next, I

> would like to ignore part of the

> SEX data (as if they were unknown), predict these missing SEX values

> and estimate the covariate

> effect of SEX on CL.

>

> Could anyone give me a hint how to accomplish this?

>

> Kind regards,

> Anthe Zandvliet

The Problem

There are two parts to this problem:

1. How can one estimate the fraction of subjects with a particular SEX e.g. female, given observed values for SEX while simultaneously  estimating PK parameters from observed values of concentration? Note that the estimation of the fraction does not require NONMEM. It is typically done more conveniently by counting the number of subjects for each sex.

2. Given the first model but now with some missing values for SEX how can a reasonable guess be made for SEX in order to improve the estimates of an influence of SEX on clearance?
Methods

There are 2 distinct methods.

Mixture Method: uses a mixture model to describe variability in clearance as bimodal based on concentration observations alone. It is then assumed that one of these modes is a predictor of SEX. Note that this method only uses concentrations and does not use observations of SEX as a dependent variable. SEX is only used as a covariate to describe variability in clearance.
Likelihood Method: involves a joint model for the likelihood of the observed concentrations and observed sex. The joint likelihood can be predicted by two variants 

A) Calculation in $ERROR

B) Calculated by NONMEM with user defined subroutines CCONTR and CONTR

These two methods (and variants of the Likelihood Method) can be combined to use information about SEX obtained from the distribution of clearance as well as using observed values of SEX to determine the most likely value for SEX when it is missing.
Runs were perfomed using NONMEM V Release 1.1 patched with NMQUAL version 4.1.1. The compiler was Intel Version 9 with -O3 /Qprec_div /Qprec /nologo options.

Data

The original data set (cpsex.csv) was used to estimate the PK parameters and the effect of female sex (SEX=0) on clearance.

Table 1 Data Sets
	Filename
	DV
	Description

	cpsex.csv
	Conc and Sex
	Original data from Anthe

	cpsex_droppedsex.csv
	Conc and Sex
	Random missing sex; Missing sex cases dropped

	cpsex_missingsex.csv
	Conc and Sex
	Random missing sex

	cp_missingsex.csv
	Conc only
	Random missing sex


The original data set was used to create data with the SEX covariate randomly missing in 50% of cases. Three kinds of datasets were created based on missing SEX (see Table 1).
Results

The fit statistics for combinations of model methods and data sets are shown in Table 2.

Table 2 Fit statistics

	Model
	SEX?
	Mix
	Joint
	Obj
	Sig
	Sub
	Obs

	cpsex_droppedsex_ccontr
	dropped
	n
	CCONTR
	356.217
	5.4
	29
	102

	cp_mixedmissingsex
	missing
	y
	na
	684.219
	5.5
	59
	155

	cpsex_mixedmissingsex_2LL
	missing
	y
	$ERROR
	724.11
	5.4
	59
	184

	cpsex_mixedmissingsex_ccontr
	missing
	y
	CCONTR
	724.11
	5.9
	59
	184

	cpsex_2LL
	complete
	n
	$ERROR
	759.198
	6.4
	59
	214

	cpsex_ccontr
	complete
	n
	CCONTR
	759.198
	5.8
	59
	214

	cpsex_mixedsex_2LL
	complete
	y
	$ERROR
	759.198
	3.7
	59
	214

	cpsex_mixedsex_ccontr
	complete
	y
	CCONTR
	759.198
	6.5
	59
	214


The objective function obtained with the CCONTR and $ERROR variants was identical but the number of significant digits tended to be lower with the $ERROR variant. This is consistent with unpublished observations of the performance of these variants (fewer successful runs, biased parameter estimates with $ERROR variant).
Table 3 Parameter Estimates
	Model
	PFEM
	POP

CL
	FFEM

CL
	FEM

CL

 Bias
	Male

CL

Bias
	PPV

CL
	POP

V
	PPV

V
	SD

	cpsex_droppedsex_ccontr
	0.448
	0.00574
	1.11
	-18%
	10%
	0.41
	1.59
	0.34
	2.46

	cp_mixedmissingsex
	0.448
	0.00604
	1.08
	-15%
	16%
	0.38
	1.56
	0.43
	2.53

	cpsex_mixedmissingsex_2LL
	0.447
	0.00604
	1.08
	-15%
	16%
	0.38
	1.56
	0.43
	2.53

	cpsex_mixedmissingsex_ccontr
	0.447
	0.00604
	1.08
	-15%
	16%
	0.38
	1.56
	0.43
	2.53

	cpsex_2LL
	0.424
	0.00522
	1.4
	0%
	0%
	0.33
	1.56
	0.43
	2.52

	cpsex_ccontr
	0.424
	0.00522
	1.4
	0%
	0%
	0.33
	1.56
	0.43
	2.52

	cpsex_mixedsex_2LL
	0.424
	0.00522
	1.4
	0%
	0%
	0.33
	1.56
	0.43
	2.52

	cpsex_mixedsex_ccontr
	0.424
	0.00522
	1.4
	0%
	0%
	0.33
	1.56
	0.43
	2.52


Discussion

Use of a mixture model alone was similar in performance to a joint likelihood model with a mixture model when SEX was missing.
A joint likelihood model without a mixture for SEX provided biased estimates of PPV and somewhat larger biased estimates of female clearance. Male clearance bias was less than the other methods.
Conclusion

A mixture model is helpful in reducing bias due to missing covariates.

Caution

These results should not be over-interpreted as a true reflection of the method properties. A simulation based study is required to assess if there are consistent differences between results obtained with these methods.

Acknowledgement – Helpful suggestions for improving this document were provided by Celine Laffont.

Mixture Method Alone

$PROB PHENOBARBITAL POPULATION PK MODEL
$DATA cp_missingsex.csv

$INPUT C ID TIME AMT WT APGR DV=Z EVID MDV SEX FLAG

$EST MAXEVAL=9999 SIG=6 PRINT=20 NOABORT

METHOD=COND LAPLACE

MSFO=mixedsex.MSF

$COV

$THETA

   (0,0.424,1) FIX ; PFEM

   (0, 0.005) ; POP_CL

   (0, 1.5)   ; POP_V

   (0,1.4) ; FFEMCL

   (0,2.5) ; SD

$OMEGA

 0.16 ; PPV_CL

 0.16 ; PPV_V

$SIGMA

 1 FIX ; EPS1

$SUBROUTINES ADVAN1

$MIX

NSPOP=2

P(1)=THETA(1)

P(2)=1-THETA(1)

$PK

   IF (SEX.GE.0) THEN ; not missing

       ISEX=SEX

   ELSE

      IF (MIXNUM.EQ.1) THEN ; female

         ISEX=0

      ELSE

         ISEX=1

      ENDIF

   ENDIF

   TVCL=THETA(2)*THETA(4)**ISEX

   CL=TVCL*EXP(ETA(1))

   TVV=THETA(3)

   V=TVV*EXP(ETA(2))

   K=CL/V

   S1=V

TAD=TIME ; for a single dose only

SID=ID

$ERROR

   IPRED=F

   SD=THETA(5)

   Y=IPRED + SD*ERR(1)

$TABLE ID SID TIME TAD IPRED CL ETA(1) SEX FLAG

       V ETA(2) WT APGR NOPRINT ONEHEADER FILE=115.TAB

Mixture and Likelihood Method ($ERROR)

$PROB PHENOBARBITAL POPULATION PK MODEL
$DATA cpsex_missingsex.csv

$INPUT C ID TIME AMT WT APGR DV=Z EVID MDV SEX FLAG

$EST MAXEVAL=9999 SIG=6 PRINT=20 NOABORT

METHOD=COND LAPLACE -2LL

MSFO=mixedsex.MSF

$COV

$THETA

   (0,0.4,1) ; PFEM

   (0, 0.005) ; POP_CL

   (0, 1.5)   ; POP_V

   (0,1.4) ; FFEMCL

   (0,2.5) ; SD

$OMEGA

 0.16 ; PPV_CL

 0.16 ; PPV_V

$SUBROUTINE ADVAN=1 TRANS1 

$MIX

NSPOP=2

P(1)=THETA(1)

P(2)=1-THETA(1)

$PK

   IF (SEX.GE.0) THEN ; not missing

       ISEX=SEX

   ELSE

      IF (MIXNUM.EQ.1) THEN ; female

         ISEX=0

      ELSE

         ISEX=1

      ENDIF

   ENDIF

   TVCL=THETA(2)*THETA(4)**ISEX

   CL=TVCL*EXP(ETA(1))

   TVV=THETA(3)

   V=TVV*EXP(ETA(2))

   K=CL/V

   S1=V

TAD=TIME ; for a single dose only

SID=ID

$ERROR

   IPRED=F

   SD=THETA(5)

   IF (FLAG.EQ.1) THEN

      Y=((DV-IPRED)/SD)**2 + 2*LOG(SD)

   ENDIF

   IF (FLAG.EQ.2.AND.DV.EQ.0) THEN

      Y=-2*LOG(THETA(1))

   ENDIF

   IF (FLAG.EQ.2.AND.DV.EQ.1) THEN

      Y=-2*LOG(1-THETA(1))

   ENDIF

$TABLE ID SID TIME TAD IPRED CL ETA(1) SEX FLAG

       V ETA(2) WT APGR NOPRINT ONEHEADER FILE=115.TAB

Mixture and Likelihood Method (CCONTR)

$PROB PHENOBARBITAL POPULATION PK MODEL
$DATA cpsex_missingsex.csv

$INPUT C ID TIME AMT WT APGR DV=Z EVID MDV SEX FLAG

$EST MAXEVAL=9999 SIG=6 PRINT=20 NOABORT

METHOD=COND LAPLACE

MSFO=mixedsex.MSF

$COV

$THETA

   (0,0.4,1) ; PFEM

   (0, 0.005) ; POP_CL

   (0, 1.5)   ; POP_V

   (0,1.4) ; FFEMCL

   (0,2.5) ; SD

$OMEGA

 0.16 ; PPV_CL

 0.16 ; PPV_V

$SIGMA

 1 FIX ; EPS1

$CONTR DATA=(FLAG)

$SUBROUTINE ADVAN=1 TRANS1 

CONTR=../contr.for

CCONTR=../ccontr.for

$MIX

NSPOP=2

P(1)=THETA(1)

P(2)=1-THETA(1)

$PK

   IF (SEX.GE.0) THEN ; not missing

       ISEX=SEX

   ELSE

      IF (MIXNUM.EQ.1) THEN ; female

         ISEX=0

      ELSE

         ISEX=1

      ENDIF

   ENDIF

   TVCL=THETA(2)*THETA(4)**ISEX

   CL=TVCL*EXP(ETA(1))

   TVV=THETA(3)

   V=TVV*EXP(ETA(2))

   K=CL/V

   S1=V

TAD=TIME ; for a single dose only

SID=ID

$ERROR (OBS)

   IPRED=F

   SD=THETA(5)

   IF (FLAG.EQ.1) THEN

      Y=IPRED + SD*ERR(1)

   ENDIF

   IF (FLAG.EQ.2.AND.DV.EQ.0) THEN

      Y=THETA(1)

   ENDIF

   IF (FLAG.EQ.2.AND.DV.EQ.1) THEN

      Y=1-THETA(1)

   ENDIF

$TABLE ID SID TIME TAD IPRED CL ETA(1) SEX FLAG

       V ETA(2) WT APGR NOPRINT ONEHEADER FILE=115.TAB

      SUBROUTINE CCONTR (ICALL,CNT,P1,P2,IER1,IER2)

      SAVE

C LVR and NO should be changed to match values in NSIZES

      PARAMETER(LVR=30,NO=50)

      COMMON /ROCM4/ Y(NO),DATA(NO,3)

      DOUBLE PRECISION CNT,P1,P2,Y

      DIMENSION P1(*),P2(LVR,*)

      TYPE=DATA(1,1)

C Value of TYPE is provided as a user defined data item

C CELS is used for continuous type data

C CLIK is used for LIKE or -2LL

      IF (TYPE.EQ.1)THEN

         CALL CELS(CNT,P1,P2,IER1,IER2)

      ELSE

C CLIK first argument 1 means LIKE and 2 means -2LL 

         CALL CLIK(1,CNT,P1,P2,IER1,IER2)

      ENDIF

      RETURN

      END

      SUBROUTINE CONTR (ICALL, CNT,IER1,IER2)

      SAVE

      DOUBLE PRECISION CNT

      CALL NCONTR (CNT,IER1,IER2,L2R)

      RETURN

      END

